
7 Exact Solution Methods

This chapter introduces a model known as a Markov decision process (MDP) to
represent sequential decision problems where the effects of our actions are uncer-
tain.1 We beginwith a description of themodel, which specifies both the stochastic

1 Such models were originally stud-
ied in the 1950s. R. E. Bellman, Dy-
namic Programming. Princeton Uni-
versity Press, 1957. Amodern treat-
ment can be found in M.L. Put-
erman, Markov Decision Processes:
Discrete Stochastic Dynamic Program-
ming. Wiley, 2005.

dynamics of the system as well as the utility associated with its evolution. Dif-
ferent algorithms can be used to compute the utility associated with a decision
strategy and to search for an optimal strategy. Under certain assumptions, we
can find exact solutions to Markov decision processes. Later chapters will discuss
approximation methods that tend to scale better to larger problems.

7.1 Markov Decision Processes

In an MDP (algorithm 7.1), we choose action at at time t based on observing state
st. We then receive a reward rt. The action spaceA is the set of possible actions, and
the state space S is the set of possible states. Some of the algorithms assume these
sets are finite, but this is not required in general. The state evolves probabilistically
based on the current state and action we take. The assumption that the next state
depends only on the current state and action and not on any prior state or action
is known as the Markov assumption.

A1 A2 A3

R1 R2 R3

S1 S2 S3

Figure 7.1. Markov decision pro-
cess decision diagram.

At

Rt

St St+1

Figure 7.2. Stationary Markov de-
cision process decision diagram.

An MDP can be represented using a decision network as shown in figure 7.1.
There are information edges (not shown in the figure) from A1:t−1 and S1:t to At.
The utility function is decomposed into rewards R1:t. We focus on stationaryMDPs
in which P(St+1 | St, At) and P(Rt | St, At) do not vary with time. Stationary
MDPs can be compactly represented by a dynamic decision diagram as shown
in figure 7.2. The state transition model T(s′ | s, a) represents the probability of
transitioning from state s to s′ after executing action a. The reward function R(s, a)
represents the expected reward received when executing action a from state s.

The reward function is a deterministic function of s and a because it represents
an expectation, but rewards may be generated stochastically in the environment
or even depend upon the resulting next state.2 Example 7.1 shows how to frame

2 For example, if the reward de-
pends on the next state as given
by R(s, a, s′), then the (expected)
reward function would be

R(s, a) = ∑
s′

T(s′ | s, a)R(s, a, s′)a collision avoidance problem as an MDP.

The problem of aircraft collision avoidance can be formulated as anMDP. The
states represent the positions and velocities of our aircraft and the intruder
aircraft, and the actions represent whether we climb, descend, or stay level.
We receive a large negative reward for colliding with the other aircraft and a
small negative reward for climbing or descending.

Given knowledge of the current state, we must decide whether an avoid-
ance maneuver is required. The problem is challenging because the positions
of the aircraft evolve probabilistically and we want to make sure that we start
our maneuver early enough to avoid collision but late enough so that we
avoid unnecessary maneuvering.

Example 7.1. Aircraft collision
avoidance framed as an MDP.

struct MDP

γ # discount factor

𝒮 # state space

𝒜 # action space

T # transition function

R # reward function

TR # sample transition and reward

end

Algorithm 7.1. Data structure for
an MDP. We will use the TR field
later to sample the next state and
reward given the current state and
action: s′, r = TR(s, a).

The rewards in anMDP are treated as components in an additively decomposed
utility function. In a finite horizon problem with n decisions, the utility associated
with a sequence of rewards r1:n is simply

n

∑
t=1

rt (7.1)

The sum of rewards is sometimes called the return.
In an infinite horizon problem in which the number of decisions is unbounded,

the sum of rewards can become infinite.3 There are several ways to define utility

3 Suppose strategy A results in a
reward of 1 per time step and strat-
egy B results in a reward of 100
per time step. Intuitively, a ratio-
nal agent should prefer strategy B
over strategy A, but both provide
the same infinite expected utility.in terms of individual rewards in infinite horizon problems. One way is to impose

a discount factor γ between 0 and 1. The utility is given by

∞

∑
t=1

γt−1rt (7.2)

This value is sometimes called the discounted return. So long as 0 ≤ γ < 1 and the
rewards are finite, the utility will be finite. The discount factor makes it so that
rewards in the present are worth more than rewards in the future, a concept that
also appears in economics.

Another way to define utility in infinite horizon problems is to use the average
reward given by

lim
n→∞

1
n

n

∑
t=1

rt (7.3)

This average return formulation can be attractive because we do not have to choose
a discount factor, but there is often not a practical difference between this for-
mulation and discounted return with a discount factor close to 1. Because the
discounted return is often computationally simpler to work with, we will focus
on the discounted formulation.

A policy tells us what action to select given the past history of states and actions.
The action to select at time t, given the history ht = (s1:t, a1:t−1), is written πt(ht).
Because the future states and rewards depend only on the current state and action
(as made apparent in the conditional independence assumptions in figure 7.1),
we can restrict our attention to policies that depend only on the current state.
In addition, we will primarily focus on deterministic policies because, in MDPs,
there is guaranteed to exist an optimal policy that is deterministic. Later chapters
discuss stochastic policies, where πt(at | st) denotes the probability the policy
assigns to taking action at in state st at time t.

In infinite horizon problems with stationary transitions and rewards, we can
further restrict our attention to stationary policies, which do not depend on time.We
will write the action associated with stationary policy π in state s as π(s), without
the temporal subscript. In finite horizon problems, however, it may be beneficial
to select a different action depending on how many time steps are remaining. For
example, when playing basketball, it is generally not a good strategy to attempt a
half-court shot unless there are only a couple seconds remaining in the game.

132 chapter 7. exact solution methods

The expected utility of executing π from state s is denoted Uπ(s). In the context
of MDPs, Uπ is often referred to as the value function. An optimal policy π∗ is a
policy that maximizes expected utility:4 4 Doing so is consistent with the

maximum expected utility princi-
ple introduced in section 6.4.π∗(s) = arg max

π
Uπ(s) (7.4)

for all states s. Depending on the model, there may be multiple policies that are
optimal. The value function associated with an optimal policy π∗ is called the
optimal value function and is denoted U∗.

An optimal policy can be found by using a computational technique called
dynamic programming,5 which involves simplifying a complicated problem by

5 The term dynamic programming
was coined by the American math-
ematician Richard Ernest Bellman
(1920–1984). Dynamic refers to
the fact that the problem is time-
varying and programming refers
to a methodology to find an opti-
mal program or decision strategy.
R. Bellman, Eye of the Hurricane:
an Autobiography. World Scientific,
1984.

breaking it down into simpler sub-problems in a recursive manner. Although
we will focus on dynamic programming algorithms for MDPs, dynamic pro-
gramming is a general technique that can be applied to a wide variety of other
problems. For example, dynamic programming can be used in computing a Fi-
bonacci sequence and finding the longest common subsequence between two
strings.6 In general, algorithms that use dynamic programming for solving MDPs 6 T.H. Cormen, C. E. Leiserson,

R. L. Rivest, and C. Stein, Intro-
duction to Algorithms, 3rd ed. MIT
Press, 2009.

are much more efficient than brute force methods.

7.2 Policy Evaluation

Before we discuss how to go about computing an optimal policy, we will first dis-
cuss policy evaluation, where we compute the value function Uπ . Policy evaluation
can be done iteratively. If the policy is executed for a single time step, the utility
is Uπ

1 (s) = R(s, π(s)). Further steps can be obtained by applying the lookahead
equation:

Uπ
k+1(s) = R(s, π(s)) + γ ∑

s′
T(s′ | s, π(s))Uπ

k (s
′) (7.5)

This equation is implemented in algorithm 7.2. Iterative policy evaluation is
implemented in algorithm 7.3. Several iterations are shown in figure 7.3.

The value function Uπ can be computed to arbitrary precision given sufficient
iterations. For an infinite horizon, we have

Uπ(s) = R(s, π(s)) + γ ∑
s′

T(s′ | s, π(s))Uπ(s′) (7.6)

function lookahead(𝒫::MDP, U, s, a)

𝒮, T, R, γ = 𝒫.𝒮, 𝒫.T, 𝒫.R, 𝒫.γ

return R(s,a) + γ*sum(T(s,a,s′)*U(s′) for s′ in 𝒮)

end

function lookahead(𝒫::MDP, U::Vector, s, a)

𝒮, T, R, γ = 𝒫.𝒮, 𝒫.T, 𝒫.R, 𝒫.γ

return R(s,a) + γ*sum(T(s,a,s′)*U[i] for (i,s′) in enumerate(𝒮))

end

Algorithm 7.2. Functions for com-
puting the lookahead state-action
value from a state s given an action
a using an estimate of the value
function U for the MDP 𝒫. The sec-
ond version handles the case when
U is a vector.

function iterative_policy_evaluation(𝒫::MDP, π, k_max)

𝒮, T, R, γ = 𝒫.𝒮, 𝒫.T, 𝒫.R, 𝒫.γ

U = [0.0 for s in 𝒮]

for k in 1:k_max

U = [lookahead(𝒫, U, s, π(s)) for s in 𝒮]

end

return U

end

Algorithm 7.3. Iterative policy
evaluation, which iteratively com-
putes the value function for a pol-
icy π for MDP 𝒫 with discrete state
and action spaces using k_max iter-
ations.

Iteration 1 Iteration 2

Iteration 3 Iteration 4

−10 −8 −6 −4 −2 0 2 4 6 8 10

Figure 7.3. Iterative policy eval-
uation used to evaluate an east-
moving policy on the hex world
problem (appendix F.1). The ar-
rows indicate the direction recom-
mended by the policy (i.e., always
move east), and the colors indi-
cate the values associated with the
states. The values changewith each
iteration.

134 chapter 7. exact solution methods

at convergence. Convergence can be proven using the fact that the update in
equation (7.5) is a contraction mapping (reviewed in appendix A.15).7 7 See exercise 7.11.

Policy evaluation can be donewithout iteration by solving a systemof equations
in matrix form:

Uπ = Rπ + γTπUπ (7.7)

where Uπ and Rπ are the utility and reward functions represented in vector
form with |S| components. The |S| × |S| matrix Tπ contains state transition
probabilities where Tπ

ij is the probability of transitioning from the ith state to the
jth state.

The value function is obtained as follows:

Uπ − γTπUπ = Rπ (7.8)
(I− γTπ)Uπ = Rπ (7.9)

Uπ = (I− γTπ)−1Rπ (7.10)

This method is implemented in algorithm 7.4. Solving for Uπ in this way
requires O(|S|3) time. The method is used to evaluate a policy in figure 7.4.

function policy_evaluation(𝒫::MDP, π)

𝒮, R, T, γ = 𝒫.𝒮, 𝒫.R, 𝒫.T, 𝒫.γ

R′ = [R(s, π(s)) for s in 𝒮]

T′ = [T(s, π(s), s′) for s in 𝒮, s′ in 𝒮]

return (I - γ*T′)\R′

end

Algorithm 7.4. Exact policy eval-
uation, which computes the value
function for a policy π for an MDP
𝒫 with discrete state and action
spaces.

Figure 7.4. Exact policy evaluation
used to evaluate an east-moving
policy for the hex world problem.
The exact solution contains lower
values than what was contained in
the first few steps of iterative pol-
icy evaluation in figure 7.3. If we
ran iterative policy evaluation for
more iterations, it would converge
to the same value function.7.3 Value Function Policies

The previous section showed how to compute a value function associated with a
policy. This section shows how to extract a policy from a value function, which
we later use when generating optimal policies. Given a value function U, which

may or may not correspond to the optimal value function, we can construct a
policy π that maximizes the lookahead equation introduced in equation (7.5):

π(s) = arg max
a

(
R(s, a) + γ ∑

s′
T(s′ | s, a)U(s′)

)
(7.11)

We refer to this policy as a greedy policy with respect to U. If U is the optimal
value function, then the extracted policy is optimal. Algorithm 7.5 implements
this idea.

An alternative way to represent a policy is to use the action value function,
sometimes called the Q-function. The action value function represents the expected
return when starting in state s, taking action a, and then continuing with the
greedy policy with respect to Q:

Q(s, a) = R(s, a) + γ ∑
s′

T(s′ | s, a)U(s′) (7.12)

From this action value function, we can obtain the value function,

U(s) = max
a

Q(s, a) (7.13)

as well as the policy,
π(s) = arg max

a
Q(s, a) (7.14)

Storing Q explicitly for discrete problems requires O(|S| × |A|) storage instead
of O(|S|) storage for U, but we do not have to use R and T to extract the policy.

Policies can also be represented using the advantage function, which quantifies
the advantage of taking an action in comparison to the greedy action. It is defined
in terms of the difference between Q and U:

A(s, a) = Q(s, a)−U(s) (7.15)

Greedy actions have zero advantage, and non-greedy actions have negative advan-
tage. Some algorithms that we will discuss later in the book use U representations,
but others will use Q or A.

struct ValueFunctionPolicy

𝒫 # problem

U # utility function

end

function greedy(𝒫::MDP, U, s)

u, a = _findmax(a->lookahead(𝒫, U, s, a), 𝒫.𝒜)

return (a=a, u=u)

end

function (π::ValueFunctionPolicy)(s)

return greedy(π.𝒫, π.U, s).a

end

Algorithm 7.5. A value function
policy extracted from a value func-
tion U for an MDP 𝒫. The greedy

function will be used in other algo-
rithms.

7.4 Policy Iteration

Policy iteration (algorithm 7.6) is one way to compute an optimal policy. It in-
volves iterating between policy evaluation (section 7.2) and policy improvement
through lookahead (algorithm 7.5). Policy iteration is guaranteed to converge
given any initial policy. It converges because there are finitely many policies and
every iteration improves the policy if it is possible to be improved. Figure 7.5
demonstrates policy iteration on the hex world problem.

struct PolicyIteration

π # initial policy

k_max # maximum number of iterations

end

function solve(M::PolicyIteration, 𝒫::MDP)

π, 𝒮 = M.π, 𝒫.𝒮

converged = false

for k = 1:M.k_max

U = policy_evaluation(𝒫, π)

π′ = ValueFunctionPolicy(𝒫, U)

if all(π(s) == π′(s) for s in 𝒮)

break

end

π = π′

end

return π

end

Algorithm 7.6. Policy iteration,
which iteratively improves an ini-
tial policy π to obtain an optimal
policy for an MDP 𝒫 with discrete
state and action spaces.

Iteration 1 Iteration 2

Iteration 3 Iteration 4

Figure 7.5. Policy iteration used to
iteratively improve an initially east-
moving policy on the hex world
problem to obtain an optimal pol-
icy. In the first iteration, we see the
value function associated with the
east-moving policy and the arrows
indicating the policy that is greedy
with respect to that value function.
Policy iteration converges in four
iterations; if we ran for a fifth itera-
tion, we would get the same policy.

Policy iteration tends to be expensive because we must evaluate the policy in
each iteration. A variation of policy iteration is called modified policy iteration,8 8 M.L. Puterman and M.C. Shin,

‘‘Modified Policy Iteration Algo-
rithms for Discounted Markov De-
cision Problems,’’ Management Sci-
ence, vol. 24, no. 11, pp. 1127–1137,
1978.

which approximates the value function using iterative policy evaluation instead
of exact policy evaluation. A parameter in the modified policy iteration algorithm
is the number of policy evaluation iterations to perform between steps of policy
improvement. If this parameter is 1, then this approach is identical to value
iteration.

7.5 Value Iteration

Value iteration is an alternative to policy iteration that is often used because of its
simplicity. Unlike policy improvement, value iteration updates the value function
directly. It begins with any bounded value function U, meaning that |U(s)| < ∞
for all s. One common initialization is U(s) = 0 for all s.

The value function can be improved by applying the Bellman equation:9 9 Named for one of the pioneers of
the field of dynamic programming.
R. E. Bellman, Dynamic Program-
ming. Princeton University Press,
1957.

Uk+1(s) = max
a

(
R(s, a) + γ ∑

s′
T(s′ | s, a)Uk(s′)

)
(7.16)

This backup procedure is implemented in algorithm 7.7.

function backup(𝒫::MDP, U, s)

return maximum(lookahead(𝒫, U, s, a) for a in 𝒫.𝒜)

end

Algorithm 7.7. The backup proce-
dure applied to an MDP 𝒫, which
improves a value function U at state
s.

Repeated application of this update is guaranteed to converge on an optimal
policy. Like iterative policy evaluation, we can use the fact that the update is a

contraction mapping to prove convergence.10 This optimal policy is guaranteed 10 See exercise 7.12.

to satisfy:

U∗(s) = max
a

(
R(s, a) + γ ∑

s′
T(s′ | s, a)U∗(s′)

)
(7.17)

Further applications of Bellman’s equation once this equality holds do not change
the policy. An optimal policy can be extracted from U∗ using equation (7.11).
Value iteration is implemented in algorithm 7.8 and is applied to the hex world
problem in figure 7.6.

The implementation in algorithm 7.8 stops after a fixed number of iterations,
but it is also common to terminate the iterations early based on the maximum
change in value called the Bellman residual, ‖Uk+1−Uk‖∞. If the Bellman residual
drops below some threshold δ, then the iterations terminate. A Bellman residual
of δ guarantees that the optimal value function obtained by value iteration is
within ε = δγ/(1− γ) of U∗.11 Discount factors closer to 1 significantly inflate 11 See exercise 7.8.

this error, leading to slower convergence. If we heavily discount future reward (γ

closer to 0), then we do not need to iterate as much into the future. This effect is
demonstrated in example 7.2.

Knowing the maximum deviation of the estimated value function from the
optimal value function, ‖Uk − U∗‖∞ < ε, allows us to bound the maximum
deviation of reward obtained under the extracted policy π from an optimal policy
π∗, the policy loss ‖Uπ −U∗‖∞. The policy loss is bounded by 2ε.12 12 To see why, see exercise 7.9.

struct ValueIteration

k_max # maximum number of iterations

end

function solve(M::ValueIteration, 𝒫::MDP)

U = [0.0 for s in 𝒫.𝒮]

for k = 1:M.k_max

U = [backup(𝒫, U, s) for s in 𝒫.𝒮]

end

return ValueFunctionPolicy(𝒫, U)

end

Algorithm 7.8. Value iteration,
which iteratively improves a value
function U to obtain an optimal pol-
icy for anMDP 𝒫with discrete state
and action spaces. The method ter-
minates after k_max iterations.

Iteration 0 Iteration 1

Iteration 2 Iteration 3

Iteration 4 Iteration 5

Iteration 6 Iteration 7

Figure 7.6. Value iteration on the
hex world problem to obtain an op-
timal policy. Each hex is colored ac-
cording to the value function and
arrows indicate the policy that is
greedy with respect to that value
function.

Consider a simple variation of the hex world problem consisting of a straight
line of tiles with a single consuming tile at the end producing 10 reward. The
discount factor directly affects the rate at which reward from the consuming
tile propagates down the line to the other tiles, and thus how quickly value
iteration converges.

γ = 0.9 γ = 0.5

Example 7.2. An example showing
the effect of the discount factor on
convergence of value iteration. In
each case value iteration was run
until the Bellman residual was less
than 1.

7.6 Asynchronous Value Iteration

Value iteration tends to be computationally intensive, as every entry in the value
function Uk is updated in each iteration to obtain Uk+1. In asynchronous value itera-
tion, only a subset of the states are updated in each iteration. Asynchronous value
iteration is still guaranteed to converge on the optimal value function provided
that each state is updated infinitely often.

One common asynchronous value iteration method, Gauss-Seidel value iteration
(algorithm 7.9), sweeps through an ordering of the states and applies the Bellman
update in place:

U(s)← max
a

(
R(s, a) + γ ∑

s′
T(s′ | s, a)U(s′)

)
(7.18)

The computational savings lies in not having to construct a second value function
in memory in each iteration. Gauss-Seidel value iteration can converge more
quickly than standard value iteration depending on the ordering chosen.13 In 13 A poor ordering in Gauss-Seidel

value iteration cannot cause the al-
gorithm to be slower than standard
value iteration.

some problems, the state contains a time index that increments deterministically
forward in time. If we apply Gauss-Seidel value iteration starting at the last time
index and work our way backwards, this process is sometimes called backwards
induction value iteration. An example of the impact of the state ordering is given in
example 7.3.

struct GaussSeidelValueIteration

k_max # maximum number of iterations

end

function solve(M::GaussSeidelValueIteration, 𝒫::MDP)

𝒮, 𝒜, T, R, γ = 𝒫.𝒮, 𝒫.𝒜, 𝒫.T, 𝒫.R, 𝒫.γ

U = [0.0 for s in 𝒮]

for k = 1:M.k_max

for (s, i) in enumerate(𝒮)

u = backup(𝒫, U, s)

U[i] = u

end

end

return ValueFunctionPolicy(𝒫, U)

end

Algorithm 7.9. Asynchronous
value iteration, which updates
states in a different manner to
value iteration, often saving com-
putation time. The method termi-
nates after k_max iterations.

Consider the linear variation of the hex world problem from example 7.2. We
can solve the same problem using asynchronous value iteration. The ordering
of the states directly affects the rate at which reward from the consuming
tile propagates down the line to the other tiles, and thus how quickly the
method converges.

Left to Right Right to Left

Example 7.3. An example show-
ing the effect of the state ordering
on convergence of asynchronous
value iteration. In this case, evalu-
ating right to left allows for conver-
gence in far fewer iterations.

7.7 Linear Program Formulation

The problem of finding an optimal policy can be formulated as a linear program,
which is an optimization problem with a linear objective function and a set of
linear equality or inequality constraints. Once a problem is represented as a linear
program, we can use one of many different linear programming solvers.14 14 For an overview of linear pro-

gramming, see R. Vanderbei, Lin-
ear Programming, Foundations and
Extensions, 4th ed. Springer, 2014.

To show how we can convert the Bellman equation into a linear program, we
begin by replacing the equality in the Bellman equation with a set of inequality
constraints while minimizing U(s) at each state s:15 15 Intuitively, we want to push the

value U(s) at all states s down to
convert the inequality constraints
into equality constraints. Hence,
we minimize the sum of all utili-
ties.

minimize ∑
s

U(s)

subject to U(s) ≥ max
a

(
R(s, a) + γ ∑

s′
T(s′ | s, a)U(s′)

)
for all s

(7.19)

The variables in the optimization are the utilities at each state. Once we know
those utilities, we can extract an optimal policy using equation (7.11).

The maximization in the inequality constraints can be replaced by a set of
linear constraints, making it a linear program:

minimize ∑
s

U(s)

subject to U(s) ≥ R(s, a) + γ ∑
s′

T(s′ | s, a)U(s′) for all s and a
(7.20)

In the linear program above, the number of variables is equal to the number
of states and the number of constraints is equal to the number of states times
the number of actions. Because linear programs can be solved in polynomial
time,16 MDPs can be solved in polynomial time. Although a linear program-

16 This was proven by L.G.
Khachiyan, ‘‘Polynomial Algo-
rithms in Linear Programming,’’
USSR Computational Mathematics
and Mathematical Physics, vol. 20,
no. 1, pp. 53–72, 1980. Modern
algorithms tend to be more
efficient in practice.

ming approach provides this asymptotic complexity guarantee, it is often more
efficient in practice to simply use value iteration. Algorithm 7.10 provides an
implementation.

7.8 Linear Systems with Quadratic Reward

So far, we have assumed discrete state and action spaces. This section relaxes
this assumption, allowing for continuous, vector-valued states and actions. The
Bellman equation for discrete problems can be modified as follows:17

17 This section assumes the prob-
lem is undiscounted and finite hori-
zon, but these equations can be eas-
ily generalized.

struct LinearProgramFormulation end

function tensorform(𝒫::MDP)

𝒮, 𝒜, R, T = 𝒫.𝒮, 𝒫.𝒜, 𝒫.R, 𝒫.T

𝒮′ = eachindex(𝒮)

𝒜′ = eachindex(𝒜)

R′ = [R(s,a) for s in 𝒮, a in 𝒜]

T′ = [T(s,a,s′) for s in 𝒮, a in 𝒜, s′ in 𝒮]

return 𝒮′, 𝒜′, R′, T′

end

solve(𝒫::MDP) = solve(LinearProgramFormulation(), 𝒫)

function solve(M::LinearProgramFormulation, 𝒫::MDP)

𝒮, 𝒜, R, T = tensorform(𝒫)

model = Model(GLPK.Optimizer)

@variable(model, U[𝒮])

@objective(model, Min, sum(U))

@constraint(model, [s=𝒮,a=𝒜], U[s] ≥ R[s,a] + 𝒫.γ*T[s,a,:]⋅U)

optimize!(model)

return ValueFunctionPolicy(𝒫, value.(U))

end

Algorithm 7.10. Linear program
formulation. For convenience in
specifying the linear program, we
define a function for converting an
MDP into its tensor form, where
the states and actions consist of
integer indices, the reward func-
tion is a matrix, and the transition
function is a three-dimensional ten-
sor. It uses the JuMP.jl package for
mathematical programming. The
optimizer is set to use GLPK.jl, but
others can be used instead. We also
define the default solve behavior
for MDPs to use this formulation.

Uh+1(s) = max
a

(
R(s, a) +

∫
T(s′ | s, a)Uh(s

′)ds′
)

(7.21)

where s and a in equation (7.16) are replaced with their vector equivalents, the
summation is replaced with an integral, and T provides a probability density
rather than a probability mass. Computing equation (7.21) is not straightforward
for an arbitrary continuous transition distribution and reward function.

In some cases, exact solution methods do exist for MDPs with continuous
state and action spaces.18 In particular, if a problem has linear dynamics and has 18 For a detailed overview, see

Chapter 4 of Volume I of D. Bert-
sekas, Dynamic Programming and
Optimal Control. Athena Scientific,
2007.

quadratic reward, then the optimal policy can be efficiently found in closed form.
Such a system is known in control theory as a linear–quadratic regulator (LQR) and
has been well studied.19

19 For a compact summary of LQR
and other related control problems,
see A. Shaiju and I. R. Petersen,
‘‘Formulas for Discrete Time LQR,
LQG, LEQG and Minimax LQG
Optimal Control Problems,’’ IFAC
Proceedings Volumes, vol. 41, no. 2,
pp. 8773–8778, 2008.

A problem has linear dynamics if the transition function has the form:

T(s′ | s, a) = Tss + Taa + w (7.22)

where Ts and Ta are matrices that determine the mean of the next state s′ given s
and a, and w is a random disturbance drawn from a zero mean, finite variance dis-
tribution that does not depend on s and a. One common choice is the multivariate
Gaussian.

A reward function is quadratic if it can be written in the form:

R(s, a) = s>Rss + a>Raa (7.23)

where Rs and Ra are matrices that determine how state and action component
combinations contribute reward. We additionally require that Rs be negative
semidefinite and Ra be negative definite. Such a reward function penalizes states
and actions that deviate from 0.

Problems with linear dynamics and quadratic reward are common in control
theory where one often seeks to regulate a process such that it does not deviate far
from adesired value. The quadratic cost assignsmuch higher cost to states far from
the origin than those near it. The optimal policy for problemwith linear dynamics
and quadratic reward has an analytic, closed-form solution. Many MDPs can be
approximated with linear–quadratic MDPs and solved, often yielding reasonable
policies for the original problem.

Substituting the transition and reward functions into equation (7.21) produces:

Uh+1(s) = max
a

(
s>Rss + a>Raa +

∫
p(w)Uh(Tss + Taa + w)dw

)
(7.24)

where p(w) is the probability density of the random, zero-mean disturbance w.
The optimal one-step value function is:

U1(s) = max
a

(
s>Rss + a>Raa

)
= s>Rss (7.25)

for which the optimal action is a = 0.
We will show through induction that Uh(s) has a quadratic form s>Vhs + qh

with symmetric matrices Vh. For the one-step value function, V1 = Rs and q1 = 0.
Substituting this quadratic form into equation (7.24) yields:

Uh+1(s) = s>Rss + max
a

(
a>Raa +

∫
p(w)

(
(Tss + Taa + w)>Vh(Tss + Taa + w) + qh

)
dw
)

(7.26)

This can be simplified by expanding and using the fact that
∫

p(w)dw = 1
and

∫
wp(w)dw = 0:

Uh+1(s) = s>Rss + s>T>s VhTss

+ max
a

(
a>Raa + 2s>T>s VhTaa + a>T>a VhTaa

)
+
∫

p(w)
(

w>Vhw
)

dw + qh

(7.27)

We can obtain the optimal action by differentiating with respect to a and setting
it to 0:20 20 Recall that:

∇xAx = A>

∇xx>Ax = (A + A>)x
0 =

(
Ra + R>a

)
a + 2T>a VhTss +

(
T>a VhTa +

(
T>a VhTa

)>)
a

= 2Raa + 2T>a VhTss + 2T>a VhTaa
(7.28)

Solving for the optimal action yields:21 21 The matrix Ra + T>a VhTa is neg-
ative definite and thus invertible.

a = −
(

Ra + T>a VhTa

)−1
T>a VhTss (7.29)

Substituting the optimal action into Uh+1(s) yields the quadratic form that we
were seeking Uh+1(s) = s>Vh+1s + qh+1, with22 22 This equation is sometimes re-

ferred to as the discrete-time Riccati
equation.

Vh+1 = Rs + T>s V>h Ts −
(

T>a VhTs

)>(
Ra + T>a VhTa

)−1(
T>a VhTs

)
(7.30)

and

qh+1 =
h

∑
i=1

Ew

[
w>Viw

]
(7.31)

If w ∼ N (0, Σ), then

qh+1 =
h

∑
i=1

Tr(ΣVi) (7.32)

We can compute Vh and qh up to any horizon h starting from V1 = Rs and
q1 = 0 and iterating using the equations above. The optimal action for an h-step
policy comes directly from equation (7.29):

πh(s) = −
(

T>a Vh−1Ta + Ra

)−1
T>a Vh−1Tss (7.33)

Note that the optimal action is independent of the zero-mean disturbance
distribution. The variance of the disturbance, however, does affect the expected
utility. Algorithm 7.11 provides an implementation. Example 7.4 demonstrates
this process on a simple problem with linear–Gaussian dynamics.

7.9 Summary

• Discrete Markov decision processes with bounded rewards can be solved
exactly through dynamic programming.

struct LinearQuadraticProblem

Ts # transition matrix with respect to state

Ta # transition matrix with respect to action

Rs # reward matrix with respect to state (negative semidefinite)

Ra # reward matrix with respect to action (negative definite)

h_max # horizon

end

function solve(𝒫::LinearQuadraticProblem)

Ts, Ta, Rs, Ra, h_max = 𝒫.Ts, 𝒫.Ta, 𝒫.Rs, 𝒫.Ra, 𝒫.h_max

V = zeros(size(Rs))

πs = Any[s -> zeros(size(Ta, 2))]

for h in 2:h_max

V = Ts'*(V - V*Ta*((Ta'*V*Ta + Ra) \ Ta'*V))*Ts + Rs

L = -(Ta'*V*Ta + Ra) \ Ta' * V * Ts

push!(πs, s -> L*s)

end

return πs

end

Algorithm 7.11. A method that
computes an optimal policy for
an h_max-step horizon MDP with
stochastic linear dynamics param-
eterized by matrices Ts and Ta and
quadratic reward parameterized
by matrices Rs and Ra. The method
returns a vector of policies where
entry h produces the optimal first
action in an h-step policy.

• Policy evaluation for such problems can be done exactly through matrix inver-
sion or approximated by an iterative algorithm.

• Policy iteration can be used to solve for optimal policies through iterating
between policy evaluation and policy improvement.

• Value iteration and asynchronous value iteration save computation by directly
iterating the value function.

• The problem of finding an optimal policy can be framed as a linear program
and solved in polynomial time.

• Continuous problems with linear transition functions and quadratic rewards
can be computed exactly.

7.10 Exercises

Exercise 7.1. Show that for an infinite sequence of constant rewards (rt = r for all t) the
infinite horizon discounted return converges to r/(1− γ).

Consider a continuous MDP where the state is composed of a scalar po-
sition and velocity s = [x, v]. Actions are scalar accelerations a that are
each executed over a time step ∆t = 1. Find an optimal 5-step policy from
s0 = [−10, 0] given a quadratic reward:

R(s, a) = −x2 − v2 − 0.5a2

such that the system tends towards rest at s = 0.
The transition dynamics are:[

x′

v′

]
=

[
x + v∆t + 1

2 a∆t2 + w1

+ v + a∆t + w2

]
=

[
1 ∆t
0 1

] [
x
v

]
+

[
0.5∆t2

∆t

]
[a] + w

where w is drawn from a zero-mean multivariate Gaussian distribution with
covariance 0.1I.

The reward matrices are Rs = −I and Ra = −[0.5].
The resulting optimal policies are:

π1(s) =
[
0 0

]
s

π2(s) =
[
−0.286 −0.857

]
s

π3(s) =
[
−0.462 −1.077

]
s

π4(s) =
[
−0.499 −1.118

]
s

π5(s) =
[
−0.504 −1.124

]
s

−12 −10 −8 −6 −4 −2 0 2 4

0

2

4

6

position x

sp
ee

d
v

Example 7.4. An example solving
a finite-horizon MDP with a linear
transition function and quadratic
reward. The figure shows the
progression of the system from
[−10, 0]. The blue contour lines
show the Gaussian distributions
over the state at each iteration.

Solution:We can prove that the infinite sequence of discounted constant rewards converges
to r/(1− γ) in the following steps

∞

∑
t=1

γt−1rt = r + γ1r + γ2r + · · ·

= r + γ
∞

∑
t=1

γt−1rt

We can move the summation to the left side and factor out (1− γ):

(1− γ)
∞

∑
t=1

γt−1r = r

∞

∑
t=1

γt−1r =
r

1− γ

Exercise 7.2. Suppose we have a Markov decision process consisting of five states s1:5 and
two actions, to stay (aS) and continue (aC). We have the following

T(si | si, aS) = 1 for i ∈ {1, 2, 3, 4}
T(si+1 | si, aC) = 1 for i ∈ {1, 2, 3, 4}

T(s5 | s5, a) = 1 for all actions a

R(si, a) = 0 for i ∈ {1, 2, 3, 5} and for all actions a

R(s4, aS) = 0

R(s4, aC) = 10

What is the discount factor γ if the optimal value U∗(s1) = 1?

Solution: The optimal value of U∗(s1) is associated with following the optimal policy π∗

starting from s1. Given the transition model, the optimal policy from s1 is to continue until
reaching s5, which is a terminal state where we can no longer transition to another state or
accumulate additional reward. Thus, the optimal value of s1 can be computed as

U∗(s1) =
∞

∑
t=1

γt−1rt

U∗(s1) = R(s1, aC) + γ1R(s2, aC) + γ2R(s3, aC) + γ3R(s4, aC) + γ4R(s5, aC) + · · ·

U∗(s1) = 0 + γ1 × 0 + γ2 × 0 + γ3 × 10 + γ4 × 0 + 0

1 = 10γ3

Thus, the discount factor is γ = 0.11/3 ≈ 0.464

Exercise 7.3. What is the time complexity of performing k steps of iterative policy evalua-
tion?

Solution: Iterative policy evaluation requires computing the lookahead equation:

Uπ
k+1(s) = R(s, π(s)) + γ ∑

s′
T(s′ | s, π(s))Uπ

k (s
′)

Updating the value at a single state requires summing over all |S| states. For a single
iteration over all states, we must do this operation |S| times. Thus, the time complexity of
k steps of iterative policy evaluation is O(k|S|2).

Exercise 7.4. Suppose we have an MDP with six states s1:6 and four actions a1:4. Using
the following tabular form of the action value function Q(s, a), compute U(s), π(s), and
A(s, a).

Q(s, a) a1 a2 a3 a4

s1 0.41 0.46 0.37 0.37
s2 0.50 0.55 0.46 0.37
s3 0.60 0.50 0.38 0.44
s4 0.41 0.50 0.33 0.41
s5 0.50 0.60 0.41 0.39
s6 0.71 0.70 0.61 0.59

Solution: We can compute U(s), π(s), and A(s, a) using the following equations

U(s) = max
a

Q(s, a) π(s) = arg max
a

Q(s, a) A(s, a) = Q(s, a)−U(s)

s U(s) π(s) A(s, a1) A(s, a2) A(s, a3) A(s, a4)

s1 0.46 a2 −0.05 0.00 −0.09 −0.09
s2 0.55 a2 −0.05 0.00 −0.09 −0.18
s3 0.60 a1 0.00 −0.10 −0.22 −0.16
s4 0.50 a2 −0.09 0.00 −0.17 −0.09
s5 0.60 a2 −0.10 0.00 −0.19 −0.21
s6 0.71 a1 0.00 −0.01 −0.10 −0.12

Exercise 7.5. Suppose we have a three-tile, straight-line hex world (appendix F.1) where
the rightmost tile is an absorbing state. When we take any action in the rightmost state,
we get a reward of 10 and we are transported to a fourth terminal state where we no
longer receive any reward. Use a discount factor of γ = 0.9. Perform a single step of policy
iteration where the initial policy π has us move east in the first tile, northeast in the second
tile, and southwest in the third tile. For the policy evaluation step, write out the transition
matrix Tπ and the reward vector Rπ and then solve the infinite horizon value function Uπ

directly using matrix inversion. For the policy improvement step, compute the updated
policy π′ by maximizing the lookahead equation.

Solution: For the policy evaluation step, we use equation (7.10), repeated below:

Uπ = (I− γTπ)−1Rπ

Forming the transition matrix Tπ and reward vector Rπ with an additional state for the
terminal state, we can solve for the infinite horizon value function Uπ

Uπ =




1 0 0 0
0 1 0 0
0 0 1 0
0 0 0 1

− (0.9)


0.3 0.7 0 0
0 0.85 0.15 0
0 0 0 1
0 0 0 1



−1 

−0.3
−0.85

10
0

 ≈


1.425
2.128

10
0


For the policy improvement step, we apply equation (7.11) using the updated value
function. The actions in the arg max term correspond to aE, aNE, aNW , aW , aSW , and aSE:

π(s1) = arg max(1.425, 0.527, 0.283, 0.283, 0.283, 0.527) = aE

π(s2) = arg max(6.575, 2.128, 0.970, 1.172, 0.970, 2.128) = aE

π(s3) = arg max(10, 10, 10, 10, 10, 10) = a∗all actions are equally desirable

Exercise 7.6. Perform two steps of value iteration to the problem in exercise 7.5, starting
with an initial value function U0(s) = 0 for all s.

Solution: We need to use the Bellman equation (equation (7.16)) to perform iterative
backups of the value function. The actions in the max term correspond to aE, aNE, aNW ,
aW , aSW , and aSE. For our first iteration, the value function is zero for all states, so we only
need to consider the reward component:

U1(s1) = max(−0.3,−0.85,−1,−1,−1,−0.85) = −0.3

U1(s2) = max(−0.3,−0.85,−0.85,−0.3,−0.85,−0.85) = −0.3

U1(s3) = max(10, 10, 10, 10, 10, 10) = 10

For the second iteration

U2(s1) = max(−0.57,−1.12,−1.27,−1.27,−1.27,−1.12) = −0.57

U2(s2) = max(5.919, 0.271,−1.12,−0.57,−1.12, 0.271) = 5.919

U2(s3) = max(10, 10, 10, 10, 10, 10) = 10

Exercise 7.7. Apply one sweep of asynchronous value iteration to the problem in exer-
cise 7.5, starting with an initial value function U0(s) = 0 for all s. Update the states from
right to left.

Solution: We use the Bellman equation (equation (7.16)) to iteratively update the value
function over each state following our ordering. The actions in the max term correspond
to aE, aNE, aNW , aW , aSW , and aSE:

U(s3) = max(10, 10, 10, 10, 10, 10) = 10

U(s2) = max(6, 0.5,−0.85,−0.3,−0.85, 0.5) = 6

U(s1) = max(3.48,−0.04,−1,−1,−1,−0.04) = 3.48

Exercise 7.8. Prove that a Bellman residual of δ guarantees that the value function obtained
by value iteration is within δγ/(1− γ) of U∗(s) at every state s.

Solution: For a given Uk, suppose we know that ‖Uk −Uk−1‖∞ < δ. We can bound the
improvement in the next iteration:

Uk+1(s)−Uk(s) = max
a

(
R(s, a) + γ ∑

s′
T(s′ | s, a)Uk(s

′)

)

−max
a

(
R(s, a) + γ ∑

s′
T(s′ | s, a)Uk−1(s

′)

)

< max
a

(
R(s, a) + γ ∑

s′
T(s′ | s, a)Uk(s

′)

)

−max
a

(
R(s, a) + γ ∑

s′
T(s′ | s, a)

(
Uk(s

′)− δ
))

= δγ

Similarly,

Uk+1(s)−Uk(s) > max
a

(
R(s, a) + γ ∑

s′
T(s′ | s, a)Uk(s

′)

)

−max
a

(
R(s, a) + γ ∑

s′
T(s′ | s, a)

(
Uk(s

′) + δ
))

= −δγ

The accumulated improvement after infinite iterations is thus bounded by:

‖U∗(s)−Uk(s)‖∞ <
∞

∑
i=1

δγi =
δγ

1− γ

A Bellman residual of δ thus guarantees that the optimal value function obtained by
value iteration is within δγ/(1− γ) of U∗.

Exercise 7.9. Prove that if ‖U−U∗‖∞ < ε for an estimate of the value function U, then the
policy π extracted from U using equation (7.11) has a policy loss bound of ‖Uπ −U∗‖∞ <

2ε.

Solution: We are given ‖U − U∗‖∞ < ε. The worst-case deviation between the policy
extracted from U and an optimal policy occurs when the policy from U has value just
under U + ε and the true optimal policy has value just above U− ε. In this case, Uπ −U∗ <
U + ε− (U − ε) = 2ε.

Exercise 7.10. Why does the optimal policy obtained in example 7.4 produce actions with
greater magnitude when the horizon is greater?

Solution: The problem in example 7.4 has quadratic reward that penalizes deviations
from the origin. The longer the horizon, the greater the negative reward is that can be
accumulated, making it more worthwhile to reach the origin sooner.

Exercise 7.11. Prove that iterative policy evaluation converges to the solution of equa-
tion (7.6).

Solution: Consider iterative policy evaluation applied to a policy π as given in equa-
tion (7.5):

Uπ
k+1(s) = R(s, π(s)) + γ ∑

s′
T(s′ | s, a)Uπ

k (s
′)

Let us define an operator Bπ and rewrite the above as Uπ
k+1 = BπUπ

k . We can show that
Bπ is a contraction mapping:

BπUπ(s) = R(s, π(s)) + γ ∑
s′

T(s′ | s, a)Uπ(s′)

= R(s, π(s)) + γ ∑
s′

T(s′ | s, a)
(
Uπ(s′)− Ûπ(s′) + Ûπ(s′)

)
= BπÛπ(s) + γ ∑

s′
T(s′ | s, a)

(
Uπ(s′)− Ûπ(s′)

)
≤ BπÛπ(s) + γ‖Uπ − Ûπ‖∞

Hence, ‖BπUπ − BπÛπ‖∞ ≤ α‖Uπ − Ûπ‖∞ for α = γ, implying that Bπ is a contraction
mapping. As discussed in appendix A.15, limt→∞ Bt

πUπ
1 converges to a unique fixed point

Uπ for which Uπ = BπUπ .

Exercise 7.12. Prove that value iteration converges to a unique solution.

Solution: The value iteration update (equation (7.16)) is:

Uk+1(s) = max
a

(
R(s, a) + γ ∑

s′
T(s′ | s, a)Uk(s

′)

)

We will denote the Bellman operator as B and rewrite an application of Bellman’s equation
asUk+1 = BUk. Aswith the previous problem, if B is a contractionmapping, then repreated
application of B to U will converge to a unique fixed point.

We can show that B is a contraction mapping:

BU(s) = max
a

(
R(s, a) + γ ∑

s′
T(s′ | s, a)U(s′)

)

= max
a

(
R(s, a) + γ ∑

s′
T(s′ | s, a)

(
U(s′)− Û(s′) + Û(s′)

))
≤ BÛ(s) + γ max

a ∑
s′

T(s′ | s, a)
(
U(s′)− Û(s′)

)
≤ BÛ(s) + α‖U − Û‖∞

for α = γ maxs maxa ∑s′ T(s′ | s, a), with 0 ≤ α < 1. Hence, ‖BU − BÛ‖∞ ≤ α‖U − Û‖,
which implies B is a contraction mapping.

Exercise 7.13. Show that the point to which value iteration converges corresponds to the
optimal value function.

Solution: Let U be the value function produced by value iteration. We want to show that
U = U∗. At convergence, we have BU = U. Let U0 be a value function that maps all states
to 0. For any policy π, it follows from the definition of Bπ that BπU0 ≤ BU0. Similarly,
Bt

πU0 ≤ BtU0. Because Bt
π∗U0 → U∗ and BtU0 → U as t → ∞, it follows that U∗ ≤ U,

which can only be the case if U = U∗.

Exercise 7.14. Suppose we have a linear–Gaussian problem with quadratic reward with
disturbance w ∼ N (0, Σ). Show that the scalar term in the utility function has the form:

qh+1 =
h

∑
i=1

Ew

[
w>Viw

]
=

h

∑
i=1

Tr(ΣVi)

You may want to use the trace trick:

x>Ax = Tr
(

x>Ax
)
= Tr

(
Axx>

)

Solution: The above equation is true if Ew

[
w>Viw

]
= Tr(ΣVi). Our derivation is:

E
w∼N (0,Σ)

[
w>Viw

]
= E

w∼N (0,Σ)

[
Tr
(

w>Viw
)]

= E
w∼N (0,Σ)

[
Tr
(

Viww>
)]

= Tr
(

E
w∼N (0,Σ)

[
Viww>

])
= Tr

(
Vi E

w∼N (0,Σ)

[
ww>

])
= Tr(ViΣ)

= Tr(ΣVi)

Exercise 7.15. What is the role of the scalar term q in the LQR optimal value function, as
given in equation (7.31):

qh+1 =
h

∑
i=1

Ew

[
w>Viw

]
Solution: A matrix M is positive definite if for all non-zero x, that x>Mx > 0. In equa-
tion (7.31), every Vi is negative semi-definite, so w>Vw ≤ 0 for all w. Thus, these q terms
are guaranteed to be non-positive. This should be expected, as in LQR problems it is
impossible to obtain positive reward, and we seek instead to minimize cost.

The q scalars are offsets in the quadratic optimal value function:

U(s) = s>Vs + q

Each q represents the baseline reward around which the s>Vs term fluctuates. We
know that V is negative definite, so s>Vs ≤ 0, and q thus represents the expected reward
that one could obtain if one were at the origin, s = 0.

